
Jによる３進法の処理－その１

３進法の基本と大学センター試験の問題を解く

西川　利男

はじめに

　今年の大学センター試験(1/21)に、3進数に関するBASICによるプログラミングの問

題が出された。問題自身は稿末にあげた。

　２進法、１６進法についてはコンピュータ・ディジタル技術との関連から最近では

教科書で取り上げられるようである。しかし、単なる知識ではなく、真に理解するた

めにはプログラミングも含めて３進法を取り上げたのは非常によかったと思う。

　本稿では、センター試験の問題を解きつつ３進法の基本とプログラミングを、さら

に３進法を利用したフラクタル・グラフィックスは別稿として述べた。

　

１．10進法、60進法、12進法、２進法、８進法、16進法、そして３進法

　人類の誕生の歴史以来、月日、季節の変わり、暦などで60進法、12進法が用いられ

たが、結局、普通の数え方としては10進法が主流となった。

　近年、ディジタルコンピュータでの機構上の利点から、２進法が使われ、それを便

利に扱うものとして８進法、16進法が使われている。

２．３進法(ternary notation)とは

３進法だろうと、その方式の基本は、別に難しいことはない。素朴に１から順に算

え上げて、３になったら、上の桁に上げるだけである。

　一般に10進表示の数を３進表示に直すにはどうすればよいか。次のように考える。

　・与えられた数を３つずつのグループに分けて、半端が出たらこれが最下位の３進

　　数である。もちろん、これは1か2である。半端が出なければ0である。

　・次には先のグループとしてまとめた数に対して３つずつのグループ分けを行い、

　　その余り、0, 1, 2の内のいずれかをとり下から２位の桁とする。

　・これをグループがなくなるまで繰り返す。

　このように書くとややこしいが、普通は次のように行って、最後の商と余りを列挙

すると３進表示が得られる。その手順は何ということもない。

　　3 ) 15 …… あまり0
　    -----
    3 )  5 …… あまり2
      -----
         1

この考え方（＝アルゴリズム）に基づいて、プログラムを作れば、普通の１０進数

から３進数への変換は行われる。

大学センター試験では、これをBASICでプログラムすることが問題として出されて

いた。問題は後述。

なお、問題では、３進法への変換を多項式を使って、プログラムするようになって

いるが、このように式を与えて行うことには私は同意できない。これでは公式を暗記

して、それで計算するのが数学である、とされかねない。式を利用し計算するのは数

学ではない。式を作ること、考え出すことが数学である、と私は思う。

ここではBASICではなく、Jにより上のアルゴリズムでプログラミングしてみた。プ

ログラムは後述。

３．Jによる3進数のプログラミング

　プログラムternにより15の 3進数表示を行う。途中経過も示す。

   tern 15

M: 5

T: 0

M: 1

T: 2 0

M: 0

T: 1 2 0

** tern:1 2 0

　逆にプログラムtern_to_decにより、3進数1 2 0を 10進表示に戻す。

   tern_to_dec 1 2 0

15

Jではプリミティブ動詞#.や#:を用いれば、次のように3進数でもごく手軽に扱える。

　例えば15の 3進数表示は次のようになる。ここで、左引数は3進数表示の桁数を表

す。

   (3 3 3) #: 15

1 2 0

  また、つぎのようにすれば、桁数を指定せず、自動的に3進数表示が得られる。

   3 #.^:(_1) 15

1 2 0

　これらをまとめて行う。10までの3進数は次のようになる。

   N =: i.11

   N

0 1 2 3 4 5 6 7 8 9 10

   (3 3 3) #: N

0 0 0

0 0 1

0 0 2

0 1 0

0 1 1

0 1 2

0 2 0

0 2 1

0 2 2

1 0 0

1 0 1

　次にセンター試験の問題として出された、3進の逆転数の問題をやってみた。

   gyakuten 15

N: 15, T: 1 2 0

gyakuten: 0 2 1

   gyakuten 10

N: 10, T: 1 0 1

gyakuten: 1 0 1 

　このように、15の 3進数の各桁の数字を逆転すると変わってしまうが、一方、10の

3進数の各桁の数字は逆転しても変わらない。これを3進の逆転数とよぶことにする。

　センター試験の最後の問題として、どういった数が3進の逆転数になるか、問われ

ていたが、Jでは簡単に次のように得られる。100までの数では以下のようになる。

   gyakutensuu 101

0 1 2 4 8 10 13 16 20 23 26 28 40 52 56 68 80 82 91 100

　これらのJのプログラム・コーディングは最後に挙げた。

４．終わりに－JとAPLにおける基底の変換

　数の基底の変換は、数学で基本の操作である。APLそしてその後継者であるJでは基

本のプリミティブ動詞として備えられている。従ってJでは2進法、8進法、16進法

の計算は、他の言語と異なりプログラムを組むことなく、ごく簡単に行える。

　もっと日常の例では時刻の時分秒の変換は次のように簡単である。

　   24 60 60 #: 10000

2 46 40

10000秒は2時間 46分 40秒である。

Jのプログラム・リスト

tfd =: 3 : '3 #. ^:(_1) y'  NB. ternary from decimal

dft =: 3 : '3 #. y'         NB. decimal from ternary

NB. decimal to ternary ============================== 

NB.  tern 50 => 1 2 1 2

tern =: 3 : 0

N =. y

M =. N

T =. ''

i =. 0

while. i < (3 ^. N)   NB. repeat while less than log N based on 3 

  do.

    wr 'M: ', ": M1 =. <. M % 3    NB. integer divide by 3

    wr 'T: ', ": T =. (3|M), T     NB. residue by 3, appended as upper digit

    if. M1 = 0 do. '** tern:', (":T) return. end.

    rd 1

    M =. M1 

    i=. i + 1

  end.

)

NB. ternary to decimal ==============================

NB.  tern_to_dec 1 2 1 2 => 50

tern_to_dec =: 3 : '+/ y * 3 ^ |. i. #y'

NB. gyakuten of ternary notation ====================

NB.    gyakuten 100

NB. 1 0 2 0 1

NB. 1 0 2 0 1

NB. 1

gyakuten =: 3 : 0

wr 'N: ', (":y), ', T: ', ": T =. 3 #.(^:_1) y

'gyakuten: ', ": TT =. |. T

)

gyaku =: 3 : 0"(0)

T =. 3 #.(^:_1) y

TT =. |. T

T -: TT

)

NB.    gyakutensuu 50

NB. 0 1 2 4 8 10 13 16 20 23 26 28 40

gyakutensuu =: 3 : '(gyaku i. y)#(i. y)'   

大学センター試験問題　数学Ⅱ・数学Ｂ　第６問

　毎日新聞（2013年 1月 21日）から

